

dátum és időkezelés, profi eszközök

ez a PDF dokumentum a margitfalvi.hu-n olvasható cikk nyomtatásra szánt változata

FOGALMAK

Dátumegységekkel végzett számítások előtt tisztáznunk kell, hogy milyen dátumegységekről van szó, naptári dátumegységekről vagy az adott dátumegység nap-számával azonos időszakokról. Vegyük például a heteket! Hány hetet tartalmaz a 2021.01.14. és 2021.01.28. közötti időszak.

Cs	P	Szo	V	H	K	Sze	Cs	P	Szo	V	H	K	Sze	Cs
14	15	16	17	18	19	20	21	22	23	24	25	26	27	28

Cs	P	Szo	V	H	K	Sze	Cs	P	Szo	V	H	K	Sze	Cs
14	15	16	17	18	19	20	21	22	23	24	25	26	27	28

A válasz: egy naptári hetet és két darab... Micsodát? Mondjuk, hét-periódust. A hét esetében nem okoz nehézséget a dátumegység nap-számának megadása, de az évek és a hónapok esetében már ez is kérdés lehet.

év-periódus nap-száma $(3*365 + 1*366) / 4 \gg 365,25$
hónap-periódus nap-száma $1*((3*28 + 1*29)/4) + 4*30 + 7*31 \gg 365,25 \quad 365,25/12 \gg 30,4375$

DÁTUMEGYSÉGEK DARABSZÁMA

A napok darabszámának megállapítása nem probléma, de ne felejtjük az utolsó és az első nap különbségéhez hozzáadni egyet! Tehát napok száma: <utolsó nap dátuma>-<első nap dátuma>+1. A további dátumegységek darabszámának megállapítása sem okozhat nehézséget!

	A	B	C	D	E	F	G	H	I
1	első	utolsó	évek		hónapok		hetek		
2	nap	nap	naptári	periódus	naptári	periódus	naptári	periódus	
3	2017.07.17	2021.09.28	3	4,20	49	50,40	218	219,14	
4									
5									

C3	naptári évek =ÉV(B3)-ÉV(A3)-1
D3	év-periódusok =KEREKÍTÉS((B3-A3+1)/365,25; 2)
E3	naptári hónapok* =(12-HÓNAP(A3))+((ÉV(B3)-ÉV(A3)-1)*12)+(HÓNAP(B3)-1)
F3	hónap-periódusok =KEREKÍTÉS((B3-A3+1)/30,4375; 2)
G3	naptári hetek** =((B3-(HÉT.NAPJA(B3; 2)))-(A3+(7-HÉT.NAPJA(A3; 2))+1))/7
H3	hét-periódusok =KEREKÍTÉS((B3-A3+1)/7; 2)

- A naptári hónapok összegét három tényező alkotja: az első nap évének naptári hónapjai, a naptári évek hónapjai és az utolsó nap évének naptári hónapjai.
- A naptári hetek darabszámát a vizsgált időszak első hétfője és utolsó vasárnapja által meghatározott részből képezzük.

A dátumegység-periódusok számát század pontossággal adtam meg.

TÖRTÉV, DÁTUMTÓLIG

A program két függvényt kínál a fenti számítások alternatívájaként. A TÖRTÉV-et és a DÁTUMTÓLIGot. A TÖRTÉV az év-periódusok számát tudja meghatározni. Első és második argumentuma az első nap és az utolsó nap dátuma. A harmadik argumentuma az év-periódus nap-számát meghatározó szám. Ez az argumentum nem kötelező.

	A	B	C	D	E	F	G	H	I	J
1	első nap	utolsó nap	harmadik argumentum							részletek
2			0	1	2	3	4			
3	1976.11.23	2014.08.31	37,772	37,772	37,768	38,319	37,795	37,769	37,769	
4										
5										

C3	=KEREKÍTÉS(TÖRTÉV(A3;B3);3)
D3	=KEREKÍTÉS(TÖRTÉV(A3;B3;0);3)
E3	=KEREKÍTÉS(TÖRTÉV(A3;B3;1);3)
F3	=KEREKÍTÉS(TÖRTÉV(A3;B3;2);3)
G3	=KEREKÍTÉS(TÖRTÉV(A3;B3;3);3)
H3	=KEREKÍTÉS(TÖRTÉV(A3;B3;4);3)
I3	=KEREKÍTÉS((B3-A3)/365,25;3)

harmadik argumentum	nap-szám
nincs megadva	amerikai, 360*
0	amerikai, 360*
1	átlagolt**
2	360
3	365
4	európai, 360*

- A háromszázhatvannapos éveket harmincnapos hónapok alkotják. Az USA és az EU előírásai a hónapképzést eltérő módon szabályozzák, erre utal a 0-ás és a 4-es módszer jelzője.
- Az 1-es módszer nap-számát az első nap éve, az utolsó nap éve és az időszak naptári évei napszámainak átlaga adja.

A harmadik argumentumot a program „alap”-nak hívja. Szerintem a „kalap” jobb lett volna! Egyszerűen jobban hangzik, mert köze neki is körülbelül annyi van a nap-számhoz, mint az alapnak.

A DÁTUMTÓLIG függvény egy konspirációs számítási eszköz: a névkiegészítő nem tartalmazza, argumentum-listája titkos. Ezért, én is csak a témába-vágó funkcióit ismertetem. Bizalmasan!

Ezzel a függvénnyel egy időszak év és hónap-periódusainak számát határozhatjuk meg, de nem a nap-számok, hanem a határnapok dátumainak alapján. Első és második argumentuma az első és az utolsó nap dátuma. A harmadik argumentuma idézőjelek között áll. Ha éveket akarunk számoltatni, akkor egy ipszilon (year), ha hónapokat, akkor egy em (month). Először egy év-periódusos példát vegyünk!

	A	B	C	D	E
1	első nap	utolsó nap	"Y"	részletek	
2	2019.01.17	2021.01.16	1	1	
3	2019.01.16	2021.01.16	2	2	
4	2019.01.15	2021.01.16	2	2	
5					
6					

C2	=DÁTUMTÓLIG(A2; B2; "Y")
D2	=ÉV(B2)-ÉV(A2)- (DÁTUM(0; HÓNAP(B2); NAP(B2))<DÁTUM(0; HÓNAP(A2); NAP(A2)))

A függvény az utolsó nap éveiből kivonja az első nap éveit és ha az első nap dátuma, az éveket figyelmen kívül hagyva, nagyobb, mint az utolsó nap dátuma, akkor levon belőle egyet. Ezt figyelhetjük meg a képen az első dátum-pár esetén: 2021-2019 » 2 és ebből le von egyet, mert január 17. nagyobb, mint január 16. A „<összehasonlítás>” képlet-rész IGAZ logikai érték esetén mínusz 1-t, HAMIS esetén 0-t ad eredményül.

	A	B	C	D	E
1	első nap	utolsó nap	"M"	részletek	
2	2020.02.17	2021.02.16	11	11	
3	2020.02.16	2021.02.16	12	12	
4	2020.03.16	2021.02.16	11	11	
5	2020.03.17	2021.02.16	10	10	
6					
7					


C2	=DÁTUMTÓLIG(A2; B2; "M")
D2	=<évek száma>*12+ HA(DÁTUM(0; HÓNAP(B2); NAP(B2))<DÁTUM(0; HÓNAP(A2); NAP(A2))); 12-HÓNAP(A2)+HÓNAP(B2)-(NAP(B2)<NAP(A2)); HÓNAP(B2)-HÓNAP(A2)-(NAP(B2)<NAP(A2)))

A hónapszám képlete két tényező összegéből áll. Az első tényezőt a teljes évek száma, amelynek kiszámítását az előző példában láttuk, és a 12 szorzata adja. A második tényező értékét egy feltétel határozza meg: kisebb-e az utolsó nap dátuma, az éveket figyelmen kívül hagyva, az első nap dátumánál. Az elágazás mindkét képlete tartalmazza a dátum napjainak összehasonlítását. A kiértékelés logikai értékét itt is számmá konvertálja a program, mint ahogy azt előző példában is láttuk.

A függvénynek van „napos” változata is, de arra nincs szükségünk, mert egy időszak nap-számát egyszerűen megállapíthatjuk a „<utolsó nap dátuma>-<első nap dátuma>+1” kifejezéssel.

DÁTUM EGYÉNI MEGJELENÍTÉSE

Saját dátum-formátumot a *Cellaformázás, Szám, Kategória Egyéni* utasítással deklarálhatunk.



A Formátumkód mezőbe kell kattintanunk, majd miután letöröltük a Normál bejegyzést, meg kell adnunk a kívánt megjelenítés kódját. Az elválasztó karakterek (./<szóköz>) közvetlenül beírhatók a kódba.

a formátum kódja	2007.03.05 megjelenítése	dátumegység megjelenítése
éé	07	évszám két karakterrel
éééé	2007	évszám négy karakterrel
h	3	hónapszám*
hh	03	hónapszám két karakterrel
hhh	márc	hónap-név rövidítve
hhhh	március	hónap-név
n	5	napszám*
nn	05	napszám két karakterrel
nnn	H	nap-név rövidítve
nnnn	hétfő	nap-név

- Megjelenítés a szükséges számú karakterrel: a tíz alattiak egy, a többiek két karakterrel.

A táblázatban felsorolt kódok, az elválasztó karaktereket is beleértve, tetszőleges sorrendben és tetszőleges összeállításban használhatók.

a formátum kódja	2007.03.05 megjelenítése
éééé. hh. nn.	2007. 03. 05.
éééé. hhh. nn. nnn	2007. márc. 05. H
hhhh nn. nnnn	március 05. hétfő
éééé-hh-nn	2007-03-05

A legfontosabb szempont a jó olvashatóság! Minden formai elemnek ezt a célt kell szolgálnia: oszlopszélesség, sormagasság, vízszintes és függőleges igazítás... A program standard dátumformátuma (2021.01.21) nem elég „szellős”, ezen mindenképpen javítanunk kell! Mindenkinek ízlése szerint.

A jó olvashatóság érdekében, ne zsúfoljuk össze egyetlen cellába az összes szükséges dátumelemet! Hozunk létre segéd-oszlopot. Hivatkozunk a dátum cellájára és adjuk meg a szükséges megjelenítési kódot vagy használjuk a SZÖVEG függvényt.

	A	B	C
1	dátum	nap	
2	1941-11-29	szombat	
3	1980-04-03	csütörtök	
4	1971-05-28	péntek	
5	1998-07-01	szerda	
6	2020-03-12	csütörtök	
7	1957-02-01	péntek	
8	2000-01-21	péntek	
9	1900-09-21	péntek	
10	2016-11-22	kedd	
11	1923-12-29	szombat	
12			
13			

B2	=SZÖVEG(A2 ; "nnnn")
----	------------------------

A megadott dátumot (első argumentum) a függvény a megadott formátumban (második argumentum) szöveggént jeleníteni meg. A megjelenítési kód idézőjelek között áll.

IDŐ EGYÉNI MEGJELENÍTÉSE

Az idő egyéni formátumkódját is a *Cellaformázás, Szám, Kategória, Egyéni* utasítással deklaráljuk. Az elválasztó karaktereket (.:/<szóköz>) közvetlenül beírhatjuk a kódba.

a formátum kódja	7:03:05 megjelenítése	időegység megjelenítése
ó	7	óraszám*
óó	07	óraszám két karakterrel
p	3	percszám*
pp	03	percszám két karakterrel
m	5	másodpercszám*
mm	05	másodpercszám két karakterrel

Megjelenítés a szükséges számú karakterrel: a tíz alattiak egy, a többiek két karakterrel.

A program huszonnégyórás idő-rendszert használ de előírhatunk tizenkétórás, napszak-jelölős formátumot is.

a formátum kódja	17:03:05 megjelenítése
ó:pp	17:03
ó:pp AM/PM	5:03 du.
[–hu-HU]AM/PM ó:pp*	du. 5:03

A napszak-jelölőt az időpont elé helyező formátum. Ne hagyjuk el a szögletes zárójeles kód-részt, mert annak hiányában, a kívánt megjelenítés (du. 5:03) csak időleges! A munkafüzet következő megnyitásakor a napszak-jelölő angolul fog megjelenni (PM 5:03).

A numerikus adatok egyéni megjelenítése lehetővé teszi számok esetén a mértékegységek, időpontok és az időtartalmak esetén az időegységek nevének megjelenítését.

a formátum kódja	17:03:05 megjelenítése
ó" óra "pp" perc"	17 óra 03 perc
p" perc "mm" mperc"	3 perc 05 mperc

A mindennapi életben gyakran felmerülő igény, hogy a napok órákban vagy az órák percekben jelenjenek meg. Ezt a feladatot is egyéni formátum deklarálásával tudjuk megoldani. Bemutatom egy példán: négy ember öt nap alatt...

	A	B	C	D	E	F
1	napok	Éva	Béla	Léna	Géza	
2	hétfő	7:00	6:30	5:30	7:30	
3	kedd	6:00	6:30	8:00	6:00	
4	szerda	6:30	5:30	5:00	6:00	
5	csütörtök	5:00	6:00	6:00	5:30	
6	péntek	6:30	5:30	7:30	6:30	
7	összesen	7:00	6:00	8:00	7:30	ó:pp
8		31:00	30:00	32:00	31:30	[ó]:pp
9						
10						

Tehát a napokat órákban akarom megjeleníteni. Ez nem probléma, hiszen a napi munkavégzés formátumát (ó:pp) a program automatikusan érvényesíti az összeg-sorra. De az eredmény láthatólag nem jó. Az összeget konstanssá alakítottam és ellenőriztem a szerkesztősorban. Ez látszik a képen.

Ha összeadjuk Léna óráit 31-et kapunk. Van még kétszer 30 perce, akkor ez újabb egy óra. Összesen 32 óra. Ha ezt elosztjuk 24-el 1 napot kapunk. Ez az érték 1900.01.01-nek felel meg. Az osztási maradék a 8 óra. A szükséges kód a „[ó]:pp” lesz. A szögletes zárójel arra utasítja a programot, hogy az óránál nagyobb dátumegységeket váltsa át órára és adja hozzá az eredeti óraszámhoz.

Hasonló történik a 3:12:24 standard megjelenítésű idővel is, miután deklaráltuk a „[p]:mm” kódot. Van 3 óránk. Ez 180 perc. Ehhez kell még hozzáadnunk az eredeti percszámot, a 12-t. Így 192:24-t kapunk.

DÁTUMOK KÉPLETEN ALAPULÓ FELTÉTELES FORMÁZÁSA

A művelet a formázandó cellák kijelölésével kezdődik. A legtöbb esetben nem csak a dátumot, hanem az őt tartalmazó sort is ki akarjuk emelni, ezért az oszlop-címkék kivételével jelöljük ki a teljes táblázatot. Ezután megnyitjuk a képlet deklarálására szolgáló parancstáblát: *Kezdőlap, Stílusok, Feltételes formázás, Új szabály...*, *A formázandó cellák kijelölése képlettel.*

A formázó-képletnek logikai értéket kell eredményeznie! Ha az eredmény IGAZ, akkor a program formázza a cellát, ha HAMIS, akkor nem. Az Excel a formázó-képletet a munkalap-képlettel megegyező módon kezeli: elhelyezi (képletesen) az aktuális cellába, majd átmásolja (képletesen) a többi kijelölt cellába is.

	A	B	C	D
1	munkaszám	határidő	árbevétel	
2	2020-078	2021-03-14	9 100 000	
3	2020-173	2021-07-31	11 500 000	
4	2020-107	2021-07-08	14 300 000	
5	2020-093	2021-01-19	10 100 000	
6	2020-176	2021-11-03	13 050 000	
7	2020-162	2021-10-13	7 550 000	
8	2020-130	2021-09-04	14 800 000	
9	2020-064	2021-05-23	11 250 000	
10	2020-079	2021-10-26	11 100 000	
11	2020-069	2021-01-19	8 550 000	
12	2020-130	2021-11-15	12 950 000	
13	2020-124	2021-01-28	14 850 000	
14	2020-180	2021-12-28	11 950 000	
15				
16				

	A	B	C
1	munkaszám	határidő	árbevétel
2	2020-078	2021-03-14	9 100 000
3	2020-173	2021-07-31	11 500 000
4	2020-107	2021-07-08	14 300 000
5	2020-093	2021-01-19	10 100 000
6	2020-176	2021-11-03	13 050 000
7	2020-162	2021-10-13	7 550 000
8	2020-130	2021-09-04	14 800 000
9	2020-064	2021-05-23	11 250 000
10	2020-079	2021-10-26	11 100 000
11	2020-069	2021-01-19	8 550 000
12	2020-130	2021-11-15	12 950 000
13	2020-124	2021-01-28	14 850 000
14	2020-180	2021-12-28	11 950 000
15			
16			

Új formázási szabály

Szabálytípus kiválasztása:

- ▶ Az összes cella formázása az értékük alapján
- ▶ Csak az adott tartalmú cellák formázása
- ▶ Csak a sorrend elején vagy végén lévő értékek formázása
- ▶ Csak az átlag feletti vagy alatti értékek formázása
- ▶ Csak az egyedi vagy az ismétlődő értékek formázása
- ▶ A formázandó cellák kijelölése képlettel

Szabály leírásának szerkesztése:

Értékek formázása, ha ez a képlet igaz:

=SB2=MA{} ↑

Minta: Nincs formátum beállítva Formátum...

OK Mégse

A képen az aktuális dátum sorát kiemelő feltételes formázás látható. A formai jegyeket a *Formátum...* vezérlőre kattintva adhatjuk meg.

Feltételes formázás tulajdonságainak módosítása illetve a formázás törlése, az érintett cellák kijelölése után, a *Feltételes formázás szabálykezelője* feliratú parancstáblán történik, amelyet a *Kezdőlap, Stílusok, Feltételes formázás, Szabályok kezelése...* utasítással jeleníthetünk meg.

Feltételes formázás szabálykezelője

Formázási szabályok a következőhöz: Aktuális kijelölés

Új szabály... Szabály szerkesztése... Szabály törlése

Szabály (ebben a sorrendben alkalmazva)	Formátum	Érvényesség	Leállítás, ha igaz
Képlet: =SB2=MA{} ↑	AaBbÖöÜüYyZz	=SA\$2:\$CS14	<input type="checkbox"/>

OK Bezárás Alkalmaz

Akkor most nézzünk néhány a formázó-képletet. A válogatás teljesen esetleges, hiszen olyan sok a variáció...

az aktuálisat megelőző dátumok	= $\$B2 < MA()$
következő hétfő	= $\$B2 = MA() + (7 - HÉT.NAPJA(MA(); 2)) + 1$
aktuális hét napjai	= $\text{ÉS}(\$B2 > MA() - HÉT.NAPJA(MA(); 2);$ $\$B2 < MA() + (8 - HÉT.NAPJA(MA(); 2)))$
előző hét napjai	= $\text{ÉS}(\$B2 > MA() - (HÉT.NAPJA(MA(); 2) + 7);$ $\$B2 < MA() - (HÉT.NAPJA(MA(); 2) - 1))$
aktuális hónap napjai	= $\text{ÉS}(\$B2 > DÁTUM(ÉV(MA()); HÓNAP(MA()); 1) - 1;$ $\$B2 < DÁTUM(ÉV(MA()); HÓNAP(MA()) + 1; 1))$
aktuális negyedév napja	= $\text{ÉS}(\$B2 > DÁTUM(ÉV(MA()); PLAFON.MAT(HÓNAP(MA()); 3) - 2; 1) - 1;$ $\$B2 < DÁTUM(ÉV(MA()); PLAFON.MAT(HÓNAP(MA()); 3) + 1; 1))$
következő félév napjai	= $\text{ÉS}(\$B2 > DÁTUM(ÉV(MA()); PLAFON.MAT(HÓNAP(MA()); 6) + 1; 1);$ $\$B2 < DÁTUM(ÉV(MA()); PLAFON.MAT(HÓNAP(MA()); 6) + 7; 1))$

Több formázó-képletben szerepel a PLAFON.MAT függvény, amely az első argumentumával meghatározott számot a második argumentumának többszörösére, felfelé kerekíti.



margitfalvi.arpad@gmail.com